// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.openzeppelin.com/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * The default value of {decimals} is 18. To select a different value for
     * {decimals} you should overload it.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless this function is
     * overridden;
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `from` to `to`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(from, to, amount);

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
            // Overflow not possible: the sum of all balances is capped by totalSupply, and the sum is preserved by
            // decrementing then incrementing.
            _balances[to] += amount;
        }

        emit Transfer(from, to, amount);

        _afterTokenTransfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        unchecked {
            // Overflow not possible: balance + amount is at most totalSupply + amount, which is checked above.
            _balances[account] += amount;
        }
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
            // Overflow not possible: amount <= accountBalance <= totalSupply.
            _totalSupply -= amount;
        }

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}
}

// File: @openzeppelin/contracts/token/ERC20/extensions/ERC20Burnable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC20/extensions/ERC20Burnable.sol)

pragma solidity ^0.8.0;

import "../ERC20.sol";
import "../../../utils/Context.sol";

/**
 * @dev Extension of {ERC20} that allows token holders to destroy both their own
 * tokens and those that they have an allowance for, in a way that can be
 * recognized off-chain (via event analysis).
 */
abstract contract ERC20Burnable is Context, ERC20 {
    /**
     * @dev Destroys `amount` tokens from the caller.
     *
     * See {ERC20-_burn}.
     */
    function burn(uint256 amount) public virtual {
        _burn(_msgSender(), amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, deducting from the caller's
     * allowance.
     *
     * See {ERC20-_burn} and {ERC20-allowance}.
     *
     * Requirements:
     *
     * - the caller must have allowance for ``accounts``'s tokens of at least
     * `amount`.
     */
    function burnFrom(address account, uint256 amount) public virtual {
        _spendAllowance(account, _msgSender(), amount);
        _burn(account, amount);
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @uniswap/v2-core/contracts/interfaces/IUniswapV2Factory.sol



pragma solidity >=0.5.0;

interface IUniswapV2Factory {
    event PairCreated(address indexed token0, address indexed token1, address pair, uint);

    function feeTo() external view returns (address);
    function feeToSetter() external view returns (address);

    function getPair(address tokenA, address tokenB) external view returns (address pair);
    function allPairs(uint) external view returns (address pair);
    function allPairsLength() external view returns (uint);

    function createPair(address tokenA, address tokenB) external returns (address pair);

    function setFeeTo(address) external;
    function setFeeToSetter(address) external;
}

// File: @uniswap/v2-periphery/contracts/interfaces/IUniswapV2Router01.sol



pragma solidity >=0.6.2;

interface IUniswapV2Router01 {
    function factory() external pure returns (address);
    function WETH() external pure returns (address);

    function addLiquidity(
        address tokenA,
        address tokenB,
        uint amountADesired,
        uint amountBDesired,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline
    ) external returns (uint amountA, uint amountB, uint liquidity);
    function addLiquidityETH(
        address token,
        uint amountTokenDesired,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external payable returns (uint amountToken, uint amountETH, uint liquidity);
    function removeLiquidity(
        address tokenA,
        address tokenB,
        uint liquidity,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline
    ) external returns (uint amountA, uint amountB);
    function removeLiquidityETH(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external returns (uint amountToken, uint amountETH);
    function removeLiquidityWithPermit(
        address tokenA,
        address tokenB,
        uint liquidity,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountA, uint amountB);
    function removeLiquidityETHWithPermit(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountToken, uint amountETH);
    function swapExactTokensForTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);
    function swapTokensForExactTokens(
        uint amountOut,
        uint amountInMax,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);
    function swapExactETHForTokens(uint amountOutMin, address[] calldata path, address to, uint deadline)
        external
        payable
        returns (uint[] memory amounts);
    function swapTokensForExactETH(uint amountOut, uint amountInMax, address[] calldata path, address to, uint deadline)
        external
        returns (uint[] memory amounts);
    function swapExactTokensForETH(uint amountIn, uint amountOutMin, address[] calldata path, address to, uint deadline)
        external
        returns (uint[] memory amounts);
    function swapETHForExactTokens(uint amountOut, address[] calldata path, address to, uint deadline)
        external
        payable
        returns (uint[] memory amounts);

    function quote(uint amountA, uint reserveA, uint reserveB) external pure returns (uint amountB);
    function getAmountOut(uint amountIn, uint reserveIn, uint reserveOut) external pure returns (uint amountOut);
    function getAmountIn(uint amountOut, uint reserveIn, uint reserveOut) external pure returns (uint amountIn);
    function getAmountsOut(uint amountIn, address[] calldata path) external view returns (uint[] memory amounts);
    function getAmountsIn(uint amountOut, address[] calldata path) external view returns (uint[] memory amounts);
}

// File: @uniswap/v2-periphery/contracts/interfaces/IUniswapV2Router02.sol



pragma solidity >=0.6.2;

import './IUniswapV2Router01.sol';

interface IUniswapV2Router02 is IUniswapV2Router01 {
    function removeLiquidityETHSupportingFeeOnTransferTokens(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external returns (uint amountETH);
    function removeLiquidityETHWithPermitSupportingFeeOnTransferTokens(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountETH);

    function swapExactTokensForTokensSupportingFeeOnTransferTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external;
    function swapExactETHForTokensSupportingFeeOnTransferTokens(
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external payable;
    function swapExactTokensForETHSupportingFeeOnTransferTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external;
}

// File: contracts/lmeow/ILmeow.sol



pragma solidity ^0.8.0;

import {IERC20, IERC20Metadata} from "@openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol";

interface ILmeow is IERC20, IERC20Metadata {}

// File: contracts/lmeow/Lmeow.sol.nolog.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import {ERC20Burnable} from "@openzeppelin/contracts/token/ERC20/extensions/ERC20Burnable.sol";
import {ERC20} from "@openzeppelin/contracts/token/ERC20/ERC20.sol";
import "@uniswap/v2-core/contracts/interfaces/IUniswapV2Factory.sol";
import "@uniswap/v2-periphery/contracts/interfaces/IUniswapV2Router02.sol";
import "./ILmeow.sol";
import "./Ownable.sol";



contract Vault is Ownable {
    IUniswapV2Router02 public immutable router;

    constructor(IUniswapV2Router02 _router) {
        router = _router;
    }

    function addLiquidityETH(
        address token,
        uint amountTokenDesired,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint ethValue
    )
        external
        onlyOwner
        returns (
            uint amountToken,
            uint amountETH,
            uint liquidity
        )
    {



        (amountToken, amountETH, liquidity) = router.addLiquidityETH{value: ethValue}(
            token,
            amountTokenDesired,
            amountTokenMin,
            amountETHMin,
            to,
            block.timestamp
        );

    }

    function recover(uint256 amount) external onlyOwner {
        msg.sender.call{value: amount}(new bytes(0));
    }

    fallback() external payable onlyOwner {



        (bool success, ) = address(router).call{value: msg.value}(msg.data);
        require(success, "forward router faild");


    }

    receive() external payable {}
}

contract Lmeow is ILmeow, Ownable {
    string public constant override name = "Lmeow Token";
    string public constant override symbol = "lmeow";
    uint8 public constant override decimals = 18;
    uint256 public override totalSupply;

    uint256 public constant initialSupply = 420690000000000 ether;
    uint256 public constant initialLPSupply = (initialSupply * 75) / 100; // 75%
    uint256 public constant marketingSupply = (initialSupply * 10) / 100; // 10%
    uint256 public constant teamSupply = (initialSupply * 10) / 100; // 10%
    uint256 public constant devSupply = initialSupply - initialLPSupply - marketingSupply - teamSupply; // 5%

    address public constant marketingAddr = 0x5D98206E4a3E3b87E6d899A47f33DEB579105b96;
    address public constant teamAddr = 0x50E943A5e3c4078211a7e0226fc1e0f648B867af;
    address public constant devAddr = 0xbA6489eFB076194EB5551a0799d70648E0149952;

    uint256 public constant ticketPrice = 3000000000000 ether;
    uint256 public constant WIN_COUNT = 3;
    uint256 public constant CYCLE_DURATION = 1 weeks;
    // limit 100 = 1%, 1=0.01% ,10=0.1% , 1000=10%, 10000=100%
    uint256 public limitFractional = 100;
    uint256 public buyTaxFractional = 1000;
    uint256 public sellTaxFractional = 1000;
    uint256 public autoLpFractional = 7500;
    uint256 public constant TEN_THOUSAND = 10000;
    address public pair;
    IUniswapV2Router02 public router;
    Vault public vault;
    Vault public lpVault;
    address public weth;
    address public recoverAddress;

    uint256 public immutable deployTime;
    uint256 public lastAddLiquidityTime;
    uint256 private _seed = 2023;
    bool _inSwap;
    bool public autoAddLiquidity = true;
    bool public autoSwapBack = true;
    bool public initialLPSupplyAdded;
    mapping(address => bool) public lockedMap;
    mapping(address => uint256) public override balanceOf;
    mapping(address => bool) public withoutLimitMap;
    mapping(address => mapping(address => uint256)) public override allowance; // allowance[owner][spender]

    mapping(uint256 => mapping(uint256 => Ticket)) public ticketMap;
    mapping(uint256 => uint256) public ticketCountMap;
    mapping(uint256 => RewardCycle) public rewardCycleMap;
    mapping(uint256 => mapping(address => uint256[])) public ticketUserMap;
    mapping(uint256 => uint256) public cycleRewardAmountMap;

    event Lock(address user, uint256 timestamp);
    event OpenCycle(address user, uint256 cycleId);
    event Stake(uint256 indexed cycleId, address indexed user, uint256 count);
    event UnStake(uint256 indexed cycleId, address indexed user, uint256 count);
    event Reward(uint256 indexed cycleId, address indexed user, uint256 amount);

    error MaxHolderLimitExceeded(address user);
    error LengthMismatch();
    error AlreadySetUp();
    error AccountLocked(address user);
    error InsufficientAllowance(address owner, address spender, uint256 amount);
    error ExceedsBalance(address owner);
    error InvalidAddress(address addr);
    error InvalidTime();
    error IllegalOperation();
    error InvalidArgs();
    error OnlyHumanCall();
    error ZeroBalance();
    error TransferETHFailed(address user, uint256 amount);

    struct Ticket {
        address user;
        bool redeemed;
    }

    struct TicketDetail {
        uint256 ticketCycleId;
        uint256 ticketId;
        address user;
        bool redeemed;
    }

    struct TicketWinner {
        uint256 ticketId;
        uint256 amount;
        bool claimed;
    }

    struct RewardCycle {
        TicketWinner[] winner;
        uint256 totalRewardEth;
        bool opened;
    }

    modifier swapping() {
        _inSwap = true;
        _;
        _inSwap = false;
    }

    modifier validRecipient(address to) {
        if (to == address(0)) revert ZeroAddress();
        _;
    }

    modifier onlyEmergency() {
        if (msg.sender != recoverAddress) revert InvalidAddress(msg.sender);
        if (block.timestamp < (deployTime + 300 days)) revert IllegalOperation();
        _;
    }

    modifier holdThreshold(
        address sender,
        address recipient,
        uint256 amount
    ) {
        if (!withoutLimitMap[recipient]) {
            if (balanceOf[recipient] + amount > (initialSupply / TEN_THOUSAND) * limitFractional) {
                revert MaxHolderLimitExceeded(recipient);
            }
        }
        _;
    }

    modifier onlyHuman() {
        if (tx.origin != msg.sender) revert OnlyHumanCall();
        _;
    }

    modifier previousCycle(uint256 cycleId) {
        if (cycleId >= currentCycleId()) revert InvalidTime();
        _;
    }

    constructor(IUniswapV2Router02 router_) {
        deployTime = block.timestamp;
        router = router_;
        IUniswapV2Factory factory = IUniswapV2Factory(router.factory());
        address weth_ = router.WETH();
        weth = weth_;
        pair = factory.createPair(weth_, address(this));
        vault = new Vault(router);
        lpVault = new Vault(router);
        address sender = msg.sender;
        recoverAddress = sender;
        withoutLimitMap[pair] = true;
        withoutLimitMap[address(this)] = true;
        withoutLimitMap[address(vault)] = true;
        withoutLimitMap[address(lpVault)] = true;
        totalSupply = initialSupply;

        allowance[address(this)][address(router_)] = type(uint256).max;
        allowance[address(vault)][address(router_)] = type(uint256).max;
        allowance[address(lpVault)][address(router_)] = type(uint256).max;
        allowance[sender][address(router_)] = type(uint256).max;

        balanceOf[address(this)] = initialLPSupply;
        balanceOf[marketingAddr] = marketingSupply;
        balanceOf[teamAddr] = teamSupply;
        balanceOf[devAddr] = devSupply;

        emit Transfer(address(0), marketingAddr, marketingSupply);
        emit Transfer(address(0), teamAddr, teamSupply);
        emit Transfer(address(0), devAddr, devSupply);
        emit Transfer(address(0), address(this), initialLPSupply);
    }

    function addInitialLP() external payable onlyOwner swapping {


        if (initialLPSupplyAdded) revert IllegalOperation();
        if (msg.value != 1 ether) revert IllegalOperation();
        router.addLiquidityETH{value: 1 ether}(address(this), initialLPSupply, 0, 0, msg.sender, block.timestamp);
        initialLPSupplyAdded = true;

    }

    function setWithoutLimit(address user) external onlyOwner {
        withoutLimitMap[user] = true;
    }

    function setWithLimit(address user) external onlyOwner {
        withoutLimitMap[user] = false;
    }

    function setPair(address pair_) external onlyOwner {
        pair = pair_;
    }

    function setTaxBuyAndSellFractional(uint256 buy, uint256 sell) external onlyOwner {
        if (buy > TEN_THOUSAND / 10) revert InvalidArgs();
        if (sell > TEN_THOUSAND / 10) revert InvalidArgs();
        buyTaxFractional = buy;
        sellTaxFractional = sell;
    }

    function setAutoLpFractional(uint256 autoLpFractional_) external onlyOwner {
        if (autoLpFractional_ > TEN_THOUSAND) revert InvalidArgs();
        autoLpFractional = autoLpFractional_;
    }

    function setautoAddLiquidity(bool isAuto) external onlyOwner {
        autoAddLiquidity = isAuto;
    }

    function setAutoSwapBack(bool isAuto) external onlyOwner {
        autoSwapBack = isAuto;
    }

    /**
     * holder limit
     * @param limit 100 = 1%, 1=0.01%
     */
    function setLimitFractional(uint256 limit) external onlyOwner {
        if (limit > TEN_THOUSAND) revert InvalidArgs();
        limitFractional = limit;
    }

    function lockAccount(address[] memory users, bool locked) external onlyOwner {
        for (uint256 i = 0; i < users.length; i++) {
            lockedMap[users[i]] = locked;
        }
    }

    function _basicTransfer(
        address from,
        address to,
        uint256 amount
    ) internal returns (bool) {

        uint256 bl = balanceOf[from];


        if (bl < amount) revert ExceedsBalance(from);
        unchecked {
            balanceOf[from] = bl - amount;
            balanceOf[to] += amount;
        }
        emit Transfer(from, to, amount);
        return true;
    }

    function _transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) internal validRecipient(recipient) holdThreshold(sender, recipient, amount) returns (bool) {



        if (lockedMap[sender]) revert AccountLocked(sender);
        if (lockedMap[recipient]) revert AccountLocked(recipient);
        if (_inSwap) {

            return _basicTransfer(sender, recipient, amount);
        } else {
            if (shouldAddLiquidity()) {

                addLiquidity();

            }
            if (shouldSwapBack()) {

                swapBack();

            }

            uint256 taxAmount;
            if (sender == pair) {
                // buy
                if (buyTaxFractional > 0) {
                    // take buy tax
                    taxAmount = (amount / TEN_THOUSAND) * buyTaxFractional;
                }
            } else if (recipient == pair) {
                // sell
                if (sellTaxFractional > 0) {
                    // take sell tax
                    taxAmount = (amount / TEN_THOUSAND) * sellTaxFractional;
                }
            }

            if (taxAmount > 0) {
                balanceOf[sender] -= taxAmount;
                uint256 autoLpAmount = (taxAmount / TEN_THOUSAND) * autoLpFractional;
                uint256 rewardAmount = taxAmount - autoLpAmount;
                if (autoLpAmount > 0) {
                    balanceOf[address(lpVault)] += autoLpAmount;
                    emit Transfer(sender, address(lpVault), autoLpAmount);
                }
                if (rewardAmount > 0) {
                    balanceOf[address(vault)] += rewardAmount;
                    emit Transfer(sender, address(vault), rewardAmount);
                }
            }
            return _basicTransfer(sender, recipient, amount - taxAmount);
        }
    }

    function transferFrom(
        address from,
        address to,
        uint256 value
    ) external override returns (bool transferResult) {

        _spendAllowance(from, msg.sender, value);

        transferResult = _transferFrom(from, to, value);

    }

    function transfer(address to, uint256 value) external override returns (bool) {

        return _transferFrom(msg.sender, to, value);
    }

    function decreaseAllowance(address spender, uint256 subtractedValue) external returns (bool) {
        if (subtractedValue >= allowance[msg.sender][spender]) {
            allowance[msg.sender][spender] = 0;
        } else {
            allowance[msg.sender][spender] -= subtractedValue;
        }
        emit Approval(msg.sender, spender, allowance[msg.sender][spender]);
        return true;
    }

    function increaseAllowance(address spender, uint256 addedValue) external returns (bool) {
        allowance[msg.sender][spender] += addedValue;
        emit Approval(msg.sender, spender, allowance[msg.sender][spender]);
        return true;
    }

    function approve(address spender, uint256 value) external override returns (bool) {

        allowance[msg.sender][spender] = value;
        emit Approval(msg.sender, spender, value);
        return true;
    }
    function _burn(address account, uint256 amount) internal virtual {
        if (account == address(0)) revert ZeroAddress();
        uint256 accountBalance = balanceOf[account];
        if (accountBalance < amount) revert ExceedsBalance(account);
        unchecked {
            balanceOf[account] = accountBalance - amount;
            // Overflow not possible: amount <= accountBalance <= totalSupply.
            totalSupply -= amount;
        }

        emit Transfer(account, address(0), amount);
    }

    function burn(uint256 amount) public virtual {
        _burn(msg.sender, amount);
    }
    function burnFrom(address account, uint256 amount) public virtual {
        _spendAllowance(account, msg.sender, amount);
        _burn(account, amount);
    }

    function _spendAllowance(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        uint256 currentAllowance = allowance[owner][spender];


        if (currentAllowance != type(uint256).max) {
            if (currentAllowance < amount) revert InsufficientAllowance(owner, spender, amount);
            unchecked {
                allowance[owner][spender] = currentAllowance - amount;
            }
        }

    }

    function swapBack() public swapping {
        uint256 amountToSwap = balanceOf[address(vault)];



        if (amountToSwap < 10000 ether) {
            return;
        }
        address[] memory path = new address[](2);
        path[0] = address(this);
        path[1] = weth;



        IUniswapV2Router02(address(vault)).swapExactTokensForETHSupportingFeeOnTransferTokens(
            amountToSwap,
            0,
            path,
            address(this),
            block.timestamp
        );

    }

    function addLiquidity() public swapping {


        uint256 autoLiquidityAmount = balanceOf[address(lpVault)];
        uint256 amountToLiquify = autoLiquidityAmount / 2;
        uint256 amountToSwap = autoLiquidityAmount - amountToLiquify;

        if (amountToSwap < 100 ether) {

            return;
        }

        address[] memory path = new address[](2);
        path[0] = address(this);
        path[1] = weth;



        IUniswapV2Router02(address(lpVault)).swapExactTokensForETHSupportingFeeOnTransferTokens(
            amountToSwap,
            0,
            path,
            address(lpVault),
            block.timestamp
        );

        uint256 amountToLiquifyEth = address(lpVault).balance;

        //
        if (amountToLiquify > 0 && amountToLiquifyEth > 0) {

            lpVault.addLiquidityETH(address(this), amountToLiquify, 0, 0, address(this), amountToLiquifyEth);
        }
        lastAddLiquidityTime = block.timestamp;
        _seed = random(0, amountToSwap);

    }

    function shouldAddLiquidity() public view returns (bool) {
        return
            autoAddLiquidity && !_inSwap && msg.sender != pair && block.timestamp >= (lastAddLiquidityTime + 2 hours);
    }

    function shouldSwapBack() public view returns (bool) {
        return autoSwapBack && !_inSwap && msg.sender != pair;
    }

    function setRecover(address recover_) external onlyOwner {
        recoverAddress = recover_;
    }

    function emergencyRecover(IERC20 token, uint256 amount) external onlyEmergency {
        token.transfer(msg.sender, amount);
    }

    function emergencyRecoverEth(uint256 amount) external onlyEmergency {
        transferETH(msg.sender, amount, false);
    }

    function emergencyRecoverVault(Vault vault_, uint256 amount) external onlyEmergency {
        vault_.recover(amount);
    }

    function balanceOfEth(address addr) public view returns (uint256) {
        return addr.balance;
    }

    function ticketCycleId(uint256 time) public view returns (uint256) {
        uint256 dt = deployTime;
        if (time < dt) revert InvalidArgs();
        unchecked {
            return (time - dt) / CYCLE_DURATION;
        }
    }

    function currentCycleId() public view returns (uint256) {
        return ticketCycleId(block.timestamp);
    }

    function getRandomSeed(address user) internal view returns (uint256) {
        return
            uint256(
                keccak256(
                    abi.encodePacked(
                        user,
                        _seed,
                        block.prevrandao,
                        block.timestamp,
                        blockhash(block.number - 1),
                        gasleft()
                    )
                )
            );
    }

    function random(uint256 min, uint256 max) internal view returns (uint256) {
        // inclusive,inclusive (don't use absolute min and max values of uint256)
        // deterministic based on seed provided
        uint256 diff = max - min + 1;
        uint256 randomVar = uint256(keccak256(abi.encodePacked(getRandomSeed(pair)))) % diff;
        return randomVar + min;
    }

    function openLottery(uint256 cycleId) external onlyHuman {


        if (cycleId >= currentCycleId()) {
            revert InvalidTime();
        }
        if (ticketUserMap[cycleId][msg.sender].length == 0) {
            revert IllegalOperation();
        }


        RewardCycle storage cycle_ = rewardCycleMap[cycleId];
        if (cycle_.opened) revert IllegalOperation();

        uint256 ticketCount = ticketCountMap[cycleId];
        if (ticketCount == 0) revert LengthMismatch();
        uint256 rewardTotal = cycleRewardAmountMap[cycleId];
        if (rewardTotal == 0) revert ZeroBalance();
        cycle_.totalRewardEth = rewardTotal;
        uint256 amount = rewardTotal / WIN_COUNT;



        if (ticketCount <= WIN_COUNT) {
            uint256 restReward = rewardTotal;
            for (uint256 i; i < ticketCount; i++) {
                cycle_.winner.push(TicketWinner({ticketId: i, amount: amount, claimed: true}));
                address winer = ticketMap[cycleId][i].user;
                if (winer == address(0)) revert ZeroAddress();
                transferETH(winer, amount, true);
                emit Reward(cycleId, winer, amount);
                restReward -= amount;
            }
            transferETH(devAddr, restReward, false);
        } else {
            uint256[] memory winTicketIdsTmp = new uint256[](WIN_COUNT);

            for (uint256 i; i < WIN_COUNT; i++) {
                uint256 lastWinTicketId;
                if (i == 0) {
                    lastWinTicketId = random(0, ticketCount - 1);
                    _seed += lastWinTicketId;
                    winTicketIdsTmp[i] = lastWinTicketId;
                } else {
                    while (true) {
                        lastWinTicketId = random(0, ticketCount - 1);
                        _seed += lastWinTicketId;

                        bool duplicate;
                        for (uint256 j; j < winTicketIdsTmp.length; j++) {
                            if (winTicketIdsTmp[j] == lastWinTicketId) {
                                duplicate = true;
                                break;
                            }
                        }
                        if (!duplicate) {
                            winTicketIdsTmp[i] = lastWinTicketId;
                            break;
                        }
                    }
                }
                cycle_.winner.push(TicketWinner({ticketId: lastWinTicketId, amount: amount, claimed: true}));
                address winer = ticketMap[cycleId][lastWinTicketId].user;
                if (winer == address(0)) revert ZeroAddress();
                transferETH(winer, amount, true);
                emit Reward(cycleId, winer, amount);
            }
        }
        cycle_.opened = true;
        emit OpenCycle(msg.sender, cycleId);

    }

    function stake(uint256 count) external onlyHuman {
        if (count == 0) revert InvalidArgs();
        _basicTransfer(msg.sender, address(this), count * ticketPrice);

        uint256 cycleId = currentCycleId();
        uint256 ticketCount = ticketCountMap[cycleId];

        for (uint256 i; i < count; i++) {
            uint256 currentTicketId = ticketCount + i;
            ticketMap[cycleId][currentTicketId] = Ticket({user: msg.sender, redeemed: false});
            ticketUserMap[cycleId][msg.sender].push(currentTicketId);
        }
        ticketCountMap[cycleId] += count;
        emit Stake(cycleId, msg.sender, count);
    }

    function unStake(uint256 cycleId) external onlyHuman previousCycle(cycleId) {
        uint256[] memory ticketIds = ticketUserMap[cycleId][msg.sender];
        if (ticketIds.length == 0) revert LengthMismatch();

        RewardCycle storage cycle_ = rewardCycleMap[cycleId];
        if (!cycle_.opened) revert IllegalOperation();

        for (uint256 i; i < ticketIds.length; i++) {
            Ticket storage ticket = ticketMap[cycleId][ticketIds[i]];
            if (ticket.user != msg.sender) revert IllegalOperation();
            if (ticket.redeemed) revert IllegalOperation();
            ticket.redeemed = true;
        }

        _basicTransfer(address(this), msg.sender, ticketIds.length * ticketPrice);
        emit UnStake(cycleId, msg.sender, ticketIds.length);
    }

    function continueStake(uint256 previousCycleId) external previousCycle(previousCycleId) {
        uint256 cycleId = currentCycleId();

        uint256[] memory ticketIds = ticketUserMap[previousCycleId][msg.sender];

        if (ticketIds.length == 0) revert LengthMismatch();
        RewardCycle storage cycle_ = rewardCycleMap[previousCycleId];

        if (!cycle_.opened) revert IllegalOperation();

        for (uint256 i; i < ticketIds.length; i++) {
            Ticket storage ticket = ticketMap[previousCycleId][ticketIds[i]];
            if (ticket.user != msg.sender) revert IllegalOperation();
            if (ticket.redeemed) revert IllegalOperation();
            ticket.redeemed = true;
        }
        emit UnStake(previousCycleId, msg.sender, ticketIds.length);

        uint256 ticketIdCount = ticketCountMap[cycleId];

        for (uint256 i; i < ticketIds.length; i++) {
            ticketMap[cycleId][ticketIdCount + i] = Ticket({user: msg.sender, redeemed: false});
        }
        ticketCountMap[cycleId] = ticketIdCount + ticketIds.length;
        emit Stake(cycleId, msg.sender, ticketIds.length);

    }

    function getAllTicketsDetail(uint256 cycleId) external view returns (TicketDetail[] memory ticketsDetail) {
        uint256 ticketCount = ticketCountMap[cycleId];
        ticketsDetail = new TicketDetail[](ticketCount);
        for (uint256 i; i < ticketCount; i++) {
            Ticket storage ticket = ticketMap[cycleId][i];
            ticketsDetail[i] = TicketDetail({
                ticketCycleId: cycleId,
                ticketId: i,
                user: ticket.user,
                redeemed: ticket.redeemed
            });
        }
    }

    function getUserTicketList(uint256 cycleId, address user) external view returns (uint256[] memory ticketIds) {
        ticketIds = ticketUserMap[cycleId][user];
    }

    function getLockedMap(address[] calldata users) external view returns (bool[] memory lockedMap_) {
        lockedMap_ = new bool[](users.length);
        for (uint256 i; i < users.length; i++) {
            lockedMap_[i] = lockedMap[users[i]];
        }
    }

    function chainTime() public view returns (uint256) {
        return block.timestamp;
    }

    function transferETH(
        address target,
        uint256 amount,
        bool isCheck
    ) internal validRecipient(target) {
        (bool success, ) = target.call{value: amount}(new bytes(0));
        if (isCheck) {
            if (!success) revert TransferETHFailed(target, amount);
        }
    }

    receive() external payable {

        cycleRewardAmountMap[currentCycleId()] += msg.value;

    }
}

// File: contracts/lmeow/Ownable.sol



pragma solidity ^0.8.0;

abstract contract Ownable {
    address public owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    error OnlyOwnerCancall(address caller);
    error ZeroAddress();

    constructor() {
        _transferOwnership(msg.sender);
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        if (owner != msg.sender) revert OnlyOwnerCancall(msg.sender);
        _;
    }

    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        if (newOwner == address(0)) revert ZeroAddress();
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = owner;
        owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}